//声明2D噪音函数2(柏林噪声)【随机作用】
vec4 permute(vec4 x) {
    return mod(((x * 34.0) + 1.0) * x, 289.0);
}
vec2 fade(vec2 t) {
    return t * t * t * (t * (t * 6.0 - 15.0) + 10.0);
}
float cnoise(vec2 P) {
    vec4 Pi = floor(P.xyxy) + vec4(0.0, 0.0, 1.0, 1.0);
    vec4 Pf = fract(P.xyxy) - vec4(0.0, 0.0, 1.0, 1.0);
    Pi = mod(Pi, 289.0); // To avoid truncation effects in permutation
    vec4 ix = Pi.xzxz;
    vec4 iy = Pi.yyww;
    vec4 fx = Pf.xzxz;
    vec4 fy = Pf.yyww;
    vec4 i = permute(permute(ix) + iy);
    vec4 gx = 2.0 * fract(i * 0.0243902439) - 1.0; // 1/41 = 0.024...
    vec4 gy = abs(gx) - 0.5;
    vec4 tx = floor(gx + 0.5);
    gx = gx - tx;
    vec2 g00 = vec2(gx.x, gy.x);
    vec2 g10 = vec2(gx.y, gy.y);
    vec2 g01 = vec2(gx.z, gy.z);
    vec2 g11 = vec2(gx.w, gy.w);
    vec4 norm = 1.79284291400159 - 0.85373472095314 * vec4(dot(g00, g00), dot(g01, g01), dot(g10, g10), dot(g11, g11));
    g00 *= norm.x;
    g01 *= norm.y;
    g10 *= norm.z;
    g11 *= norm.w;
    float n00 = dot(g00, vec2(fx.x, fy.x));
    float n10 = dot(g10, vec2(fx.y, fy.y));
    float n01 = dot(g01, vec2(fx.z, fy.z));
    float n11 = dot(g11, vec2(fx.w, fy.w));
    vec2 fade_xy = fade(Pf.xy);
    vec2 n_x = mix(vec2(n00, n01), vec2(n10, n11), fade_xy.x);
    float n_xy = mix(n_x.x, n_x.y, fade_xy.y);
    return 2.3 * n_xy;
}

//传入参数
uniform float uWaresFrequency;//xz方向频率
uniform float uScale;//y方向比例
uniform float uNoiseFrequency;//噪声频率
uniform float uNoiseScale;//噪声高度比例
uniform float uXzScale;//x/z的比例
uniform float uTime;//时间
uniform float uXspeed;
uniform float uZspeed;
uniform float uNoiseSpeed;

varying float vElevation;//y坐标

void main() {

    vec4 g_Position = vec4(position, 1.0);
    vec4 w_Position = modelMatrix * g_Position;

    float elevation = sin(w_Position.x * uWaresFrequency + uTime * uXspeed) * sin(w_Position.z * uWaresFrequency * uXzScale + uTime * uZspeed);
    //噪音函数随机调整高度,绝对值再取反，尖锐
    elevation += -abs(cnoise(vec2(w_Position.xz * uNoiseFrequency + uTime * uNoiseSpeed))) * uNoiseScale;

    elevation *= uScale;

    vElevation = elevation;

    w_Position.y = elevation;

    gl_Position = projectionMatrix * viewMatrix * w_Position;
}